

Ein einfaches SPICE-Modell für den pyroelektrischen Sensor LIM-262-ZH

Herbert Eichele
herbert.eichele@th-nuernberg.de

Kurzfassung

Der pyroelektrische Sensor LIM-262-ZH¹ ist für die nicht-dispersive Infrarot-Gasanalyse zur Bestimmung der CO₂-Konzentration geeignet. Dieser Artikel beschreibt ein für den Schaltungsentwickler nützliches SPICE Makromodell. Frequenzgang und transientes Verhalten werden hinreichend genau modelliert. Anpassung der Modellparameter ermöglicht die Simulation von anderen pyroelektrischen Sensoren. Die Infrarot-Lichtintensität wird als elektrische Stromstärke abgebildet.

1. Einführung

Pyroelektrische Sensoren registrieren Wärmestrahlung und stellen dem Geräteentwickler ein Signal als elektrischen Strom oder elektrische Spannung zur Weiterverarbeitung zur Verfügung. Als wärmeempfindliche Elemente dienen Einkristalle wie z.B. Litium-Tantalat (LiTaO₃). Solche Kristalle sind piezoelektrischen oder ferroelektrischen Kristallen verwandt, deren elektrische Polarisation durch Veränderungen der Kristallstruktur beeinflusst wird.

Eintreffende Infrarotstrahlung der Strahlungsleistung P_L erwärmt den pyroelektrischen Kristall. Die Temperaturerhöhung ΔT führt über die uniaxiale Wärmeausdehnung zu Polarisationsladungen Q_P an den gegenüberliegenden Kristalloberflächen bzw. den dort angebrachten Elektroden und in der Folge zu einer messbaren elektrischen Spannung bzw. einem messbaren elektrischen Strom. Nach kurzer Zeit ist die Polarisati-




Bild 1: Gemessene Reaktion eines Kanals des Pyrosensors LIM-262-ZH auf eine gepulste Infrarotstrahlung. Die 4mV-Offset-Spannung ist abgezogen.

onsladung ausgeglichen; das Sensorsignal zeigt deshalb lediglich Temperaturänderungen an (s. Bild 1).

Der Pyrosensor LIM-262-ZH enthält in einem TO39 Gehäuse zwei Sensoren mit jeweils einem optischen Transmissionsfilterfenster für die wellenlängenspezifische Detektion von Infrarotstrahlung. ($\lambda=3,95\mu\text{m}$ (Referenzwellenlänge) bzw. $\lambda=4,27\mu\text{m}$ (CO₂-Absorptionswellenlänge)).




Bild 2: LIM-262-ZH im Schnitt¹.

2. Das Modell im Überblick

Bild 3 zeigt das Schaltplansymbol für beide Kanäle des LIM-262-ZH. Der CO₂-sensitive Messkanal (Kanal 1 mit den Anschlüssen 1 und 2) und der Referenzkanal (Kanal 2 mit den Anschlüssen 3 und 4)




Bild 3: Schaltplansymbol

sind gleich modelliert. Der Unterschied liegt ausschließlich im optischen Filter. Skizziert sind die Pyrosensoren und ein Operationsverstärker mit dem Rückkopplungswiderstand R_B zur Strom-Spannungs-Konversion. Die Eingänge 1 und 3 akzeptieren Infrarotstrahlung in der Form von Stromeinprägungen gemäß der Relation 1A Stromstärke je Watt Lichtleistung im jeweiligen Transmissionsbereich der optischen Filter ($\lambda=3,95\mu\text{m}$ mit Bandbreite 90nm bzw. $\lambda=4,27\mu\text{m}$ mit Bandbreite 170nm).

Bild 4 illustriert das Simulationsprinzip am Beispiel eines Pyrokanals. Die Stromquelle IRpwr modelliert eine gepulste Infrarotstrahlung der Leistung 2,7 μW im jeweiligen Transmissionsbereich der optischen Filter.




Bild 4: Simulationsprinzip am Beispiel eines Pyrokanals. Die Stromquelle IRpwr modelliert eine gepulste Infrarotstrahlung der Leistung 2,7 μW .




Bild 5: Vergleich von gemessenem und simuliertem Verhalten unter Verwendung des vorgestellten Modells.

3. Einzelheiten des Modells

Das Modell bildet das Arbeitsprinzip des realen Bauelements mit SPICE – Primitiven nach. Es ist demnach ein funktionsäquivalentes Makromodell aus einem thermischen und einem elektrischen Teilmodell (s. Bild 6).

$I_{infrarot}$ repräsentiert die absorbierte Infrarotstrahlungsleistung, die von der thermischen Kapazität C_{th} des pyroelektrischen Kristalls zeitlich integriert wird. Die zusammen mit dem thermischen Widerstand R_{th} definierte thermische Zeitkonstante $TAU_{th} = C_{th}R_{th}$ bildet die Integrationszeit und bestimmt damit die Temperaturerhöhung $\Delta T = T - T_{umg}$. Die Knotenspannung $V(T)$ in Volt entspricht der Temperatur T in °C. Da das Signal des pyroelektrischen Kristalls auf Temperaturänderungen reagiert, wird mit dem Hochpass C_2/R_2 näherungsweise die zeitliche Änderung dT/dt ermittelt und durch die Knotenspannung $V(dT/dt)$ dargestellt.

Die pyroelektrische Polarisationsladung Q_p wird durch Integration des Stroms der spannungsgesteuerten




Bild 7: Frequenz- und Phasengang des LIM-262-ZH Modells simuliert mit einer AC-Amplitude von $1\mu A$.

Stromquelle G_{pyro} über die Kapazität C_{pyro} des pyroelektrischen Kristalls abgebildet. Über U_1 , R_{fb} und C_{fb} wird aus dem pyroelektrischen Strom dQ_p/dt die Sensor-Ausgangsspannung erzeugt.

G_{pyro} wird durch die Knotenspannung $V(dT/dt)$ und somit von der Temperaturänderung gesteuert. Als Stromverstärkungsfaktor wird $PyroParm$ (pyroelektrischer Konversionsparameter; $[PyroParm] = Cb/K$) verwendet.

4. Modell-Code und Modell-Parameter

Das Modell ist als `.subckt LIM262ZH` im Anhang beigefügt. Aufruf-Parameter sind TAU_{th} und TAU_{el} , letzterer als Vielfaches von TAU_{th} . Die weiteren Parameter sind in `.subckt LIM262THsingle` aufgeführt und erläutert. Zahlenwerte wurden durch Vergleich eigener Messungen mit Simulationen bestimmt (s. Bild 5) bzw. aus öffentlich zugänglichen Herstellerinformationen¹ abgeleitet. Alle Simulationen wurden mit LTspice IV 4.20p² unter Linux-Ubuntu³ 12.04 / Wine⁴ 1.6.1 durchgeführt.




Bild 6: Aus SPICE – Primitiven erstelltes funktionsäquivalentes Makromodell des Pyrosensors LIM-262-ZH bestehend aus einem thermischen und einem elektrischen Teilmodell (Funktionsbeschreibung s. Text Abschnitt 3).

5. Danksagung

InfraTec¹ GmbH wird für die Bereitstellung von Testmustern gedankt.

6. Anhang

Modell-Code-Listing

```
* +++
* © H. Eichele
* Technische Hochschule Nürnberg - Georg-Simon-Ohm
* SS2014
*
* Spice Modell für den
* InfraTec pyroelektrischen Infrarot Sensor LIM-262-ZH (www.infratec.de)
*
* Diese Software darf für nicht-kommerzielle Zwecke frei verwendet werden;
* eine Verwendungs-Eignung wird weder zugesichert noch werden Haftungen
* für Folgen jeglicher Art, insbesondere Schadensfolgen, die sich aus dem
* Einsatz dieser Software ergeben könnten, übernommen.
*
*****
.subckt LIM262ZH 1 2 3 4 Params: TAUth=73ms TAUel=12*TAUth
*
*      | | | |
*      | | | | \- Ausgang elektrisch Kanal 2
*      | | | | \-- Infrarot-Eingang Kanal 2 (1A = 1W-Lichtleistung)
*      | | | | \----- Ausgang elektrisch Kanal 1
*      | | | | \----- Infrarot-Eingang Kanal 1 (1A = 1W-Lichtleistung)
*
XU1 1 2 LIM262ZHsingle TAUth={TAUth} TAUel={TAUel}
XU2 3 4 LIM262ZHsingle TAUth={TAUth} TAUel={TAUel}
*
.ends LIM262ZH
*****
.subckt LIM262ZHsingle 1 2 Params:
*
*      | |
*      | | \- Ausgang elektrisch Kanal 1
*      | | \-- Infrarot-Eingang Kanal 1 (1A = 1W-Lichtleistung)
*
* Parameter angepasst an eigene Messungen ergänzt durch Daten in den Artikeln
* www.infratec.de/fileadmin/downloads/pdf/TechPaper__Fast_Response_Dualcolor_Detector_extern.pdf
* www.infratec.de/fileadmin/media/Sensorik/pdf/Apl_Notes/Application_Detector_Basics.pdf
*
+ TAUth=73ms ; thermische Zeitkonstante (angepasst an eigene Messungen)
+ TAUel=12*TAUth ; default elektrische Zeitkonstante (angepasst an eigene Messungen)
+ TAUel={TAUel} ; elektrische Zeitkonstante (Parameter im Symbol einstellbar)
+ Rth=14k ; thermischer Widerstand [Rth]=K/W
+ Cth={TAUth/Rth} ; thermische Kapazität angepasst an TAUth
+ Tump=27 ; Umgebungstemperatur [Tump]=°C
+ PyroParm=7e-10 ; pyroelektrischer Konversionsparameter [PyroParm]=Cb/K
+ Cpyro=62p ; Kapazität des pyroelektrischen Elements
+ Rfb={TAUel/Cpyro} ; Arbeitswiderstand angepasst an elektrische Zeitkonstante
+ Cfb={Cpyro/100} ; parasitäre Rückkopplungskapazität
*
R0 1 T 1 ; Hilfswiderstand am "Lichteintritt"
*
Rth T N2 {Rth} ; thermisches Modell
Vtump N2 0 {Tump} ; V(T) = Temperatur, 1V=1°C
Cth T 0 {Cth}
*
C2 T dT/dt {0.01*Cth} ; V(dT/dt) = zeitliche Temperaturänderung
R2 dT/dt 0 {100*Rth}
*
Gpyro N1 0 0 dT/dt {PyroParm} ; el. Strom ~ zeitliche Temperaturänderung
Cpyro N1 0 {Cpyro}
Rfb N1 2 {Rfb}
Cfb N1 2 {Cfb}
XU1 N1 0 2 opamp ; Ausgangsspannung ~ el. Strom
*
.ends LIM262ZHsingle
*
***** Fortsetzung auf Folgeseite *****
```

```

***** Fortsetzung von Vorgängerseite *****

*
.subckt opamp 1 2 3 Params:
*      | | |
*      | | '- Output
*      Inputs - +
*
+ Aol=100K      ; open loop gain
+ GBW=1Meg     ; transit freq.
+ Lp=8V        ; output + voltage clamp
+ Lm=-8V       ; output - voltage clamp
*
G1 0 3 2 1 {Aol}
R3 3 0 1.
C3 3 0 {Aol/GBW/6.283}
D1 3 0 Clamp+
D2 0 3 Clamp-
*
.model clamp+ D(Vfwd={Lp} Ron=1.E-9)
.model clamp- D(Vfwd={-Lm} Ron=1.E-9)
*
.ends OpAmp

***** Ende des Modell-Codes *****

```

-
- [1] www.infratec.de (InfraTec GmbH, Infrarotsensorik und Messtechnik, Gostritzer Straße 61-63, 01217 Dresden)
 - [2] www.linear.com (Linear Technology Corporation)
 - [3] www.ubuntu.com
 - [4] www.winehq.org